OFICINA

Integrais triplas: uma abordagem computacional com o Maple

Pereira, Luiz Claudio[1]

Resumo:*Nesta oficina, com duração de 4 horas, a partir de problemas propostos, desenvolver-se-á, de forma interativa, com o uso do Maple, a resolução de integrais triplas por meio do Teorema de Fubini. Primeiro, com o Maple, construir-se-á o sólido $\Omega \subset \mathbb{R}^3$ referido no problema proposto. Após, examinar-se-á a situação, decidindo que projeção é mais conveniente. Por fim, instruir-se-á o programa a efetuar ordenadamente as integrais repetidas pertinentes. Espera-se que a presente oficina, embora de caráter elementar, estimule e incentive os participantes a usar o Maple como ferramenta de auxílio no processo de aprendizagem; particularmente, no estudo de integrais triplas.*

Palavras-chave:*Maple, integrais, triplas, teorema, Fubini.*

1 A integral tripla e o Teorema de Fubini

A construção da representação de uma situação do espaço tridimensional numa folha de papel, antes, explorada, por exemplo, em disciplinas de desenho geométrico, é uma habilidade cada vez menos desenvolvida ao longo dos anos de estudo que precedem ao ingresso no ensino superior.

Uma consequência, dentre outras, dessa inabilidade é que ao lidar com um sólido $\Omega \subset \mathbb{R}^3$, poucos são os indivíduos aptos a esboçar, com razoável acurácia, numa folha de papel, a superfície $S = \partial\Omega$ e menos ainda aqueles capazes de visualizar a projeção de Ω sobre os planos coordenados.

Essa deficiência, por sua vez, tende a repercutir gravosamente na capacidade do indivíduo em resolver problemas envolvendo integrais triplas porquanto, pelo Teorema

[1]Docente da Universidade de Brasília.

de Fubini, tem-se que

$$\iiint_\Omega f(x,y,z)\ \mathrm{d}V = \iint_U \left[\int_{\phi(x,y)}^{\psi(x,y)} f(x,y,z)\ \mathrm{d}z\right] \mathrm{d}A,$$

onde $U \subset \mathbb{R}^2$ é a projeção de Ω sobre o plano coordenado xy.

A projeção acima referida pode ser realizada sobre o plano coordenado xz ou sobre o plano coordenado yz, alterando-se, *mutatis mutandis*, a expressão dada pelo Teorema de Fubini. Nesse sentido, a partir de uma projeção sobre o plano coordenado yz, decorre que

$$\iiint_\Omega f(x,y,z)\ \mathrm{d}V = \iint_D \left[\int_{\lambda(y,z)}^{\sigma(y,z)} f(x,y,z)\ \mathrm{d}x\right] \mathrm{d}A,$$

onde $D \subset \mathbb{R}^2$ é a projeção de $\Omega \subset \mathbb{R}^3$ sobre o plano coordenado yz.

A integral dupla, de acordo com o Teorema de Fubini, é resolvida através de duas integrais repetidas, mediante análise da curva $C = \partial D$.

Nesse caso, as duas alternativas são

$$\iint_D \left[\int_{\lambda(y,z)}^{\sigma(y,z)} f(x,y,z)\ \mathrm{d}x\right] \mathrm{d}A \quad = \quad \int_a^b \left\{\int_{\alpha(y)}^{\beta(y)} \left[\int_{\lambda(y,z)}^{\sigma(y,z)} f(x,y,z)\ \mathrm{d}x\right] \mathrm{d}z\right\} \mathrm{d}y,$$

quando $C = \{(y,z) \in \mathbb{R}^2 : \alpha(y) \le z \le \beta(y) \text{ e } a \le y \le b\}$, ou

$$\iint_D \left[\int_{\lambda(y,z)}^{\sigma(y,z)} f(x,y,z)\ \mathrm{d}x\right] \mathrm{d}A \quad = \quad \int_c^d \left\{\int_{\nu(z)}^{\phi(z)} \left[\int_{\lambda(y,z)}^{\sigma(y,z)} f(x,y,z)\ \mathrm{d}x\right] \mathrm{d}y\right\} \mathrm{d}z,$$

quando $C = \{(y,z) \in \mathbb{R}^2 : \nu(z) \le y \le \phi(z) \text{ e } c \le z \le d\}$.

2 A abordagem computacional com o Maple

Conforme visto, a resolução de uma integral tripla exige que o indivíduo reconheça e decida que projeção merece ser utilizada no desenvolvimento da integral dupla. Essa, por sua vez, requer que se examine a curva que determina a fronteira da projeção. O sólido $\Omega \subset \mathbb{R}^3$ é, em geral, determinado pela superfície $S = \partial\Omega$.

Na abordagem computacional com o Maple, proposta nesta oficina, considerar-se-á que a superfície $S = \partial\Omega$ é formada pela união finita de superfícies parametrizáveis. Isso significa que existe um número finito de aplicações

$$\begin{array}{rccl} X: & D \subset \mathbb{R}^2 & \longrightarrow & \mathbb{R}^3 \\ & (u,v) & \mapsto & (x(u,v), y(u,v), z(u,v)) = X(u,v) \end{array}$$

tais que o traço de X é igual a S.

Assim, obtida uma parametrização apropriada para $\partial\Omega = S$, para exibição da superfície que determina o sólido, no Maple, usa-se o comando *plot3d* de sintaxe

$$\mathit{plot3d}\ \ (\ \ [\ \ x(u,v),\ \ y(u,v),\ \ z(u,v)\ \ ],\ \ u = a_1..a_2,\ \ v = b_1..b_2,\ \ \mathit{options}\ \ );$$

com a_1, a_2, b_1, b_2 escolhidos adequadamente.



Fig. 1: Autoria própria

Acima, exibe-se um sólido $\Omega \subset \mathbb{R}^3$, cuja superfície $S = \partial\Omega$ é formada por um cone, um cilindro e três planos, gerado pelo Maple a partir do comando *plot3d* e outros correlatos.

A partir da figura, o interessado pode tentar esboçar no papel a figura, treinando sua habilidade motora e desenvolvendo sua percepção geométrica. Demais, a manipulação deste sólido no Maple permite inferir que projeção merece ser utilizada no cálculo da integral tripla em Ω, de uma função real de três variáveis f, segundo o Teorema de Fubini.

Nesse caso, as três integrais repetidas podem ser determinadas pelo Maple por meio de três comandos *Int* encaixados, de acordo com a sintaxe

$$Int \;\; (Int \;\; (Int \;\; (\;\; f(x, y, z), \quad x = a_1..a_2, \quad y = b_1..b_2, \quad z = c_1..c_2 \;\;) \;\;) \;\;);$$

onde $a_1..a_2$, $b_1..b_2$ e $c_1..c_2$ denotam, respectivamente, a variação de x, y e z, conforme as escolhas da projeção U do sólido e da curva $C = \partial U$.

3 Problemas propostos

3.1 O volume de um prisma

Considere um prisma $\Omega \subset \mathbb{R}^3$ cuja base é o triângulo no plano xy limitado pelo eixo x e pelas retas $y = x$ e $x = 1$ e cujo topo está no plano $z = 2 - y$.

Note que $\partial\Omega = S$ é formada pela união de partes de planos.

Ora, qualquer plano $\pi \subset \mathbb{R}^3$ pode ser caracterizado por um ponto $P \in \pi$ e por dois vetores linearmente independentes $\vec{\alpha}, \vec{\beta} \in \pi$.

Nessas condições, uma parametrização *natural* de π é dada por

$$X(u, v) = P + u \cdot \vec{\alpha} + v \cdot \vec{\beta}.$$

O domínio $U \subset \mathbb{R}^2$ de X e o ponto *base* P devem ser convenientemente escolhidos de forma a retratar o plano π de acordo com a região de interesse.

Para ilustrar esse aspecto, considere o plano

$$z = 2 - y \Leftrightarrow 0 \cdot x + 1 \cdot y + 1 \cdot z = 2.$$

Tome $P = (0,0,2)$ por ponto *base*. Porque $A = (1,1,1)$ e $B = (0,2,0)$ são pontos desse plano,

$$\vec{\alpha} = \overrightarrow{PA} = (1,1,-1) \quad \text{e} \quad \vec{\beta} = \overrightarrow{PB} = (0,2,-2),$$

são vetores do plano. Por isso, uma parametrização *natural* do plano é dada por

$$X(u,v) = (u, u+2v, 2-u-2v).$$

Tomando

$$U = \left\{(u,v) \in \mathbb{R}^2 : -2 \leq u \leq 3 \text{ e } -3 \leq v \leq 4\right\},$$

a sequência de comandos no Maple

$$\begin{aligned}
&with(plots) : with(plottools) : with(Student) : with(ColorTools) :\\
&X := (u,v)->[u, u+2*v, 2-u-2*v];\\
&plot3d(X(u,v), u=-2..3, v=-3..4, scaling=constrained, transparency=0.1,\\
&colour=u-v, lightmodel=light3, style=patchnogrid, axes=normal,\\
&view=[-1..3, -1..6, -1..4], tickmarks=[2,4,3], orientation=[50,67,10]);
\end{aligned}$$

gera, para o plano $z = 2-y$, a figura abaixo.



Fig. 2: Autoria própria

Por outro lado, tomando

$$U = \left\{(u,v) \in \mathbb{R}^2 : \frac{1}{2} \leq u^2 + v^2 \leq 5\right\},$$

a sequência de comandos no Maple

$$\begin{aligned}
&with(plots) : with(plottools) : with(Student) : with(ColorTools) :\\
&X := (u,v)->[u, u+2*v, 2-u-2*v];\\
&u := (r,theta)->r*cos(theta); v := (r,theta)->r*sin(theta);\\
&X(u(r,theta), v(r,theta));\\
&plot3d(X(u(r,theta), v(r,theta)), r=sqrt(0.5)..sqrt(5), theta=0..2*Pi,\\
&scaling=unconstrained, transparency=0.2, axes=normal,\\
&colour=r*theta, lightmodel=light4, style=patchnogrid,\\
&tickmarks=[3,4,3], orientation=[50,67,10]);
\end{aligned}$$

gera, para o plano $z = 2-y$, a figura seguinte.



Fig. 3: Autoria própria

Conforme mostram as duas figuras anteriores, a escolha do domínio $U \subset \mathbb{R}^2$, da parametrização X, permite visualizar diferentes regiões da superfície $S = X(U)$.

Feitas essas considerações, retornemos ao prisma $\Omega \subset \mathbb{R}^3$ cuja base é o triângulo no plano xy limitado pelo eixo x e pelas retas $y = x$ e $x = 1$ e cujo topo está no plano $z = 2 - y$.

Nesse caso, a parametrização

$$X(u, v) = (u, u + 2v, 2 - u - 2v),$$

do plano $z = 2 - y$, é tal que $x = u \in [0, 1]$ e $y = u + 2v$, com

$$0 \leq y \leq x \quad \Leftrightarrow \quad 0 \leq u + 2v \leq u \quad \Leftrightarrow \quad -\frac{u}{2} \leq v \leq 0.$$

Disso decorre que o topo do prisma é obtido, no Maple, pela sequência de comandos

$$\begin{aligned}
&with(plots) : with(plottools) : with(Student) : with(ColorTools) :\\
&X := (u, v) - > [u, u + 2 * v, 2 - u - 2 * v];\\
&plot3d(X(u, v), u = 0..1, v = -u/2..0, scaling = constrained,\\
&transparency = 0.2, colour = \text{"}Nautical\ \ GrayViolet\text{"}, axes = none);
\end{aligned}$$

Raciocinando de modo similar, parametrizam-se os demais planos que compõem a superfície do prisma.

O programa pode então ser instruído com os comandos abaixo

$with(plots) : with(plottools) : with(Student) : with(ColorTools) :$
$\#$ *Do triângulo no plano xy, formado pelos vetores* $(1, 0, 0)$ *e* $(0, 1, 0)$ *e que passa pelo ponto* $(0, 0, 0)$.
$X[1] := (u, v) - > [u, v, 0]; X[1](u, v);$
$plnxy := plot3d(X[1](u, v), u = 0..1, v = 0..u, scaling = constrained,$
$transparency = 0.2, colour = yellow, axes = normal) :$
$\#$ *Do plano* $x = 1$, *formado pelos vetores* $(0, 1, 0)$ *e* $(0, 0, 1)$ *e que passa pelo ponto* $(1, 0, 0)$.

$X[2] := (u, v) -> [1, u, v]; X[2](u, v);$
$plnXequals1 := plot3d(X[2](u, v), u = 0..1, v = 0..2 - u, scaling = constrained,$
$transparency = 0.2, colour = "NiagaraDarkOrchid", axes = normal) :$
$\#$ *Do plano* $y = x$, *formado pelos vetores* $(0, 0, 1)$ *e* $(1, 1, 0)$ *e que passa pelo ponto* $(0, 0, 0)$.
$X[3] := (u, v) -> [u, u, v]; X[3](u, v);$
$plnYequalsX := plot3d(X[3](u, v), u = 0..1, v = 0..2 - u, scaling = constrained,$
$transparency = 0.2, colour = green, axes = normal) :$
$\#$ *Do plano* xz, *definido pelo eixo* x, *formado pelos vetores* $(1, 0, 0)$ *e* $(0, 0, 1)$ *e que passa pelo ponto* $(0, 0, 0)$.
$X[4] := (u, v) -> [u, 0, v]; X[4](u, v);$
$plnxz := plot3d(X[4](u, v), u = 0..1, v = 0..2, scaling = constrained,$
$transparency = 0.2, colour = blue, axes = normal) :$
$\#$*Plano* $z = 2 - y$ *em coordenadas cartesianas.*
$X := (u, v) -> [u, u + 2 * v, 2 - u - 2 * v]; X(u, v);$
$plnUp := plot3d(X(u, v), u = 0..1, v = -u/2..0, scaling = constrained,$
$transparency = 0.1, colour = "NauticalGrayViolet", axes = normal) :$
$\#$*Destacando apropriadamente os eixos coordenados.*
$zeixoz := arrow([0, 0, -1/3], [0, 0, 2.3], 0.025, 0.05, 0.0125,$
$cylindrical\_arrow, color = black);$
$yeixoy := arrow([0, -1/3, 0], [0, 1.3, 0], 0.025, 0.05, 0.0125,$
$cylindrical\_arrow, color = black);$
$xeixox := arrow([-1/3, 0, 0], [1.6, 0, 0], 0.025, 0.05, 0.0125,$
$cylindrical\_arrow, color = black);$
$\#$*Modelando o sólido.*
$display(\{plnUp, plnxy, plnxz, xeixox, yeixoy, zeixoz, plnXequals1, plnYequalsX\},$
$scaling = constrained, axes = normal, tickmarks = [[1], [1], [2]],$
$orientation = [21, 65, -8]);$

gerando o modelo a seguir.



Fig. 4: Autoria própria



Fig. 5: Autoria própria

Exame dinâmico, no Maple, da projeção do sólido sobre os planos coordenados revela que *não* é melhor opção a escolha da projeção sobre o plano xz, uma vez que ela é formada pela justaposição de duas regiões, conforme ilustrado abaixo, o que obrigaria à separação da integral múltipla na soma de duas integrais.



Fig. 6: Autoria própria

A projeção do sólido sobre o plano coordenado yz produz a região

$$U_{yz} = \{(y,z) \in \mathbb{R}^2 : 0 \leq z \leq 2 - y \text{ e } 0 \leq y \leq 1\}.$$

Pelo Teorema de Fubini, o volume do sólido é dado por

$$\iiint_\Omega 1 \cdot \mathrm{d}V \quad = \quad \iint_{U_{yz}} \left[\int_y^1 1 \cdot \mathrm{d}x\right] \, \mathrm{d}A \quad = \quad \int_0^1 \int_0^{2-y} \left[\int_y^1 1 \cdot \mathrm{d}x\right] \, \mathrm{d}z \; \mathrm{d}y \quad = \quad \frac{5}{6}.$$

Por seu turno, a projeção do sólido sobre o plano coordenado xy produz a região

$$U_{xy} = \{(x,y) \in \mathbb{R}^2 : 0 \leq y \leq x \text{ e } 0 \leq x \leq 1\}.$$

Pelo Teorema de Fubini, o volume do sólido é dado por

$$\iiint_\Omega 1 \cdot \mathrm{d}V \quad = \quad \iint_{U_{xy}} \left[\int_0^{2-y} 1 \cdot \mathrm{d}z\right] \, \mathrm{d}A \quad = \quad \int_0^1 \int_0^{x} \left[\int_0^{2-y} 1 \cdot \mathrm{d}z\right] \, \mathrm{d}y \; \mathrm{d}x \quad = \quad \frac{5}{6}.$$

No Maple, as integrais iteradas acima são calculadas usando a seguinte sequência de comandos:

```
# Projeção no plano coordenado yz.
# O volume do sólido é dado por
Int(1, x = y..1);
Int(Int(1, x = y..1), z = 0..2 − y);
Int(Int(Int(1, x = y..1), z = 0..2 − y), y = 0..1);
int(int(int(1, x = y..1), z = 0..2 − y), y = 0..1);
# Projeção no plano coordenado xy.
# O volume do sólido é dado por
Int(1, z = 0..2 − y);
Int(Int(1, z = 0..2 − y), y = 0..x);
Int(Int(Int(1, z = 0..2 − y), y = 0..x), x = 0..1);
int(int(int(1, z = 0..2 − y), y = 0..x), x = 0..1);
```

Uma vez que os limites de integração das integrais repetidas do prisma $\Omega \subset \mathbb{R}^3$ estão fixados pela escolha de U_{yz} ou de U_{xy}, a integral tripla de uma função qualquer $f(x,y,z)$

em Ω é, de acordo com o Teorema de Fubini, dada por

$$\begin{aligned}\iiint_\Omega f(x,y,z)\ \mathrm{d}V &= \int_0^1 \int_0^{2-y} \left[\int_y^1 f(x,y,z)\cdot \mathrm{d}x\right]\ \mathrm{d}z\ \mathrm{d}y \\ &= \int_0^1 \int_0^x \left[\int_0^{2-y} f(x,y,z)\cdot \mathrm{d}z\right]\ \mathrm{d}y\ \mathrm{d}x.\end{aligned}$$

Do ponto de vista computacional, definida a função f, o cálculo da integral tripla é efetuado imediatamente pelo programa.

A título de ilustração, considere a função $f(x,y,z) = x^2 - y^2 + z$.

No Maple, a sequência de comandos abaixo determina a integral tripla de f em Ω.

$$\begin{aligned}&\#\ Definindo\ a\ função\ dada.\\ &f := (x,y,z)->x^(2)-y^(2)+z; f(x,y,z);\\ &\#\ Verificando\ a\ integral\ tripla\ de\ forma\ inerte.\\ &Int(Int(Int(f(x,y,z), x=y..1), z=0..2-y), y=0..1);\\ &\#\ Efetuando\ a\ integral\ tripla.\\ &int(int(int(f(x,y,z), x=y..1), z=0..2-y), y=0..1);\\ &\#\ Verificando\ a\ integral\ tripla\ de\ forma\ inerte.\\ &Int(Int(Int(f(x,y,z), z=0..2-y), y=0..x), x=0..1);\\ &\#\ Efetuando\ a\ integral\ tripla.\\ &int(int(int(f(x,y,z), z=0..2-y), y=0..x), x=0..1);\end{aligned}$$

3.2 O volume de uma cunha

Considere uma cunha $\Omega \subset \mathbb{R}^3$ contida no cilindro $x^2+z^2=9$ e delimitada pelos planos $y=-2$ e $y+z=1$.

Inicialmente, note que, porque

$$(3\sin u)^2 + (3\cos u)^2 = 9(\sin^2 u + \cos^2 u) = 9,$$

uma parametrização *natural* do cilindro é dada por

$$Y_1(u,v) = (3\sin u, v, 3\cos u), \quad u \in [0, 2\pi]\,, \quad v \in \mathbb{R}.$$

O plano

$$y = -2 \quad \Leftrightarrow \quad 0\cdot x + 1\cdot y + 0\cdot z = -2,$$

passa pelo ponto *base* $(0,-2,0)$. Além disso, os vetores $\vec{\alpha} = (1,0,0)$ e $\vec{\beta} = (0,0,1)$ pertencem ao plano. Por isso, uma parametrização *natural* do plano é

$$Y_2(u,v) = (u,-2,v), \quad u\in\mathbb{R}, \quad v \in \mathbb{R}.$$

O plano

$$y+z=1 \quad \Leftrightarrow \quad z = 1-y,$$

pode ser *naturalmente* parametrizado fazendo $x=u$, $y=v$ e $z=1-v$. Noutros termos, basta tomar a aplicação

$$Y_3(u,v) = (u,v,1-v), \quad u\in\mathbb{R}, \quad v\in\mathbb{R}.$$

Evidentemente, se a construção da parametrização fosse realizada conforme o procedimento explicado anteriormente, outra aplicação, diferente de Y_3, seria obtida.

Isso mostra que *não* é única a parametrização de uma superfície.

No Maple, o modelo abaixo



Fig. 7: Autoria própria

é construído pela sequência de comandos

$restart;$
$with(plots) : with(plottools) : with(Student) : with(ColorTools) :$
$\#\ Do\ cilindro\ com\ eixo\ de\ simetria\ coincidente\ com\ o\ eixo\ y.$
$Y[1] := (u, v)- > [3 * sin(u), v, 3 * cos(u)]; Y[1](u, v);$
$cilin1 := plot3d(Y[1](u, v), u = 0..2 * Pi, v = -4..6, scaling = constrained,$
$transparency = 0.3, axes = normal, style = surfacewireframe, colour = u,$
$tickmarks = [[-2, 2], [-4, -3, 3, 4], [-3, -2, 2, 3]], orientation = [55, 61, 4]) :$
$\#\ Do\ plano\ y = -2.$
$Y[2] := (u, v)- > [u, -2, v]; Y[2](u, v);$
$pln2 := plot3d(Y[2](u, v), u = -3.5..3.5, v = -3.5..3.5, scaling = constrained,$
$transparency = 0.2, colour = yellow, axes = normal,$
$tickmarks = [[-3, -2, -1, 0, 1, 2, 3], [-3, -2, -1, 0, 1, 2, 3], [-3, -2, -1, 0, 1, 2, 3]],$
$orientation = [55, 61, 4]) :$
$\#\ Do\ plano\ z + y = 1.$
$Y[3] := (u, v)- > [u, v, 1 - v]; Y[3](u, v);$
$pln3 := plot3d(Y[3](u, v), u = -4..5, v = -4..7, scaling = constrained,$
$transparency = 0., colour = blue, axes = normal,$
$tickmarks = [[-3, 0, 1, 3], [-3, 0, 1, 2, 3], [-3, 0, 1, 2, 3]],$
$orientation = [-112, 4, 74]) :$
$\#\ Modelando\ o\ sólido.$
$display(\{cilin1, pln2, pln3\}, scaling = constrained, axes = normal,$
$tickmarks = [[-3, 0, 3], [-4, 0, 5], [-3, 0, 3]], orientation = [-112, 4, 74]);$

Os excessos apresentados no modelo podem ser suprimidos pela escolha apropriada do domínio $U \subset \mathbb{R}^2$ de cada parametrização, com eventual mudança de variáveis.

Com efeito, para o plano $y = -2$, toma-se $u = r\cos t$, $v = r\sin t$, com $t \in [0, 2\pi]$ e $r \in [0, 3]$. Nessas condições, a parametrização assume a forma

$$Y_2(r, t) = (r\cos t, -2, r\sin t)\,, \quad \text{com} \quad (r, t) \in [0, 3] \times [0, 2\pi]\,.$$

O cilindro $x^2 + y^2 = 9$ e o plano $y + z = 1$ intersectam-se, gerando uma curva C.

Da parametrização do cilindro

$$Y_1(u, v) = (3 \sin u, v, 3 \cos u),$$

decorre que $x = 3 \sin u$, $z = 3 \cos u$ e, da equação do plano, que $y = 1 - 3 \cos u$.

Por isso, uma parametrização de C é dada por

$$\sigma(u) = (3 \sin u, 1 - 3 \cos u, 3 \cos u)\,, \quad \text{com} \quad u \in [0, 2\pi]\,.$$

Em consequência, os excessos do cilindro podem ser removidos usando

$$Y_1(u, v) = (3 \sin u, v, 3 \cos u), \quad \text{com} \quad (u, v) \in [0, 2\pi] \times [-2, 1 - 3 \cos u]\,,$$

e os excessos do plano $y + z = 1$ podem ser retirados utilizando a parametrização

$$Y_3(r, t) = (r \sin t, 1 - r \cos t, r \cos t), \quad \text{com} \quad (r, t) \in [0, 3] \times [0, 2\pi]\,.$$

No Maple, a sequência de comandos,

```
# Removendo os excessos.
# Do encontro do plano y + z = 1 com o cilindro. Curva C.
sigma := u-> [3*sin(u), 1-3*cos(u), 3*cos(u)]; sigma(u);
crv := plot3d(sigma(u), u = 0..2*Pi, colour = black, thickness = 2) :
# Do cilindro com eixo de simetria coincidente com o eixo y.
cilin1 := plot3d(Y[1](u, v), u = 0..2*Pi, v = -2..1-3*cos(u),
scaling = constrained, transparency = 0.3, axes = normal,
style = surfacewireframe, colour = u,
tickmarks = [[-2, 2], [-4, -3, 3, 4], [-3, -2, 2, 3]], orientation = [55, 61, 4]) :
# Do plano y = -2.
Y[2] := (r, t)-> [r*cos(t), -2, r*sin(t)]; Y[2](r, t);
pln2 := plot3d(Y[2](r, t), r = 0..3, t = 0..2*Pi, scaling = constrained,
transparency = 0.2, colour = yellow, axes = normal,
tickmarks = [[-3, -2, -1, 0, 1, 2, 3], [-3, -2, -1, 0, 1, 2, 3], [-3, -2, -1, 0, 1, 2, 3]],
orientation = [55, 61, 4]) :
# Do plano y + z = 1.
Y[3] := (r, t)-> [r*sin(t), 1-r*cos(t), r*cos(t)]; Y[3](r, t);
pln3 := plot3d(Y[3](r, t), r = 0..3, t = 0..2*Pi, scaling = constrained,
transparency = 0.2, colour = blue, axes = normal,
tickmarks = [[-3, 0, 1, 3], [-3, 0, 1, 2, 3], [-3, 0, 1, 2, 3]],
orientation = [55, 61, 4]) :
# Modelando o sólido.
# Destacando apropriadamente os eixos coordenados.
zeixoz := arrow([0, 0, -3.5], [0, 0, 3.5], 0.025, 0.05, 0.0125,
cylindrical_arrow, color = black) :
yeixoy := arrow([0, -25/10, 0], [0, 4.5, 0], 0.025, 0.05, 0.0125,
cylindrical_arrow, color = black) :
xeixox := arrow([-3.5, 0, 0], [3.5, 0, 0], 0.025, 0.05, 0.0125,
cylindrical_arrow, color = black) :
display(cilin1, crv, pln2, pln3, xeixox, yeixoy, zeixoz, scaling = constrained,
axes = normal, tickmarks = [[-3, 0, 3], [-2, 0, 4], [-3, 0, 3]],
orientation = [-112, 4, 74]);
```

apara os excessos e gera os modelos abaixo.



Fig. 8: Autoria própria



Fig. 9: Autoria própria

Exame dinâmico, no Maple, da projeção do sólido sobre os planos coordenados revela que *não* é melhor opção a escolha da projeção sobre o plano xy, uma vez que ela é formada pela justaposição de três regiões, o que obrigaria à separação da integral múltipla na soma de três integrais.



Fig. 10: Autoria própria



Fig. 11: Autoria própria

Seja $U_{yz} \subset \mathbb{R}^2$ a projeção do sólido sobre o plano $x = 0$.

Nesse plano coordenado yz, o plano $y + z = 1$ intersecta o cilindro $x^2 + z^2 = 9$ nos pontos $(0, -2, 3)$ e $(0, 4, -3)$.

A reta que passa por esses pontos é tal que

$$x = 0 \quad \text{e} \quad \frac{-3-3}{4-(-2)} = \frac{y-4}{z-(-3)} \quad \Leftrightarrow \quad y = -z + 1.$$

Por isso, tem-se que

$$U_{yz} = \{(y, z) \in\subset \mathbb{R}^2 : -2 \leq y \leq -z + 1 \text{ e } -3 \leq z \leq 3\}.$$

Pelo Teorema de Fubini, o volume do sólido é dado por

$$\begin{aligned}
\iiint_{\Omega} 1 \cdot \mathrm{d}V &= \iint_{U_{yz}} \left[\int_{-\sqrt{9-z^2}}^{\sqrt{9-z^2}} 1 \cdot \mathrm{d}x \right] \mathrm{d}A \\
&= \int_{-3}^{3} \int_{-2}^{-z+1} \left[\int_{-\sqrt{9-z^2}}^{\sqrt{9-z^2}} 1 \cdot \mathrm{d}x \right] \mathrm{d}y \ \mathrm{d}z \\
&= 27\pi.
\end{aligned}$$

Agora, seja $U_{xz} \subset \mathbb{R}^2$ a projeção do sólido sobre o plano $y = 0$.



Fig. 12: Autoria própria



Fig. 13: Autoria própria

Por definição, em coordenadas cartesianas, tem-se

$$\begin{aligned} U_{xz} &= \{(x,z) \in \mathbb{R}^2 : 0 \leq x^2 + z^2 \leq 9\} \\ &= \left\{(x,z) \in \mathbb{R}^2 : -\sqrt{9-z^2} \leq x \leq \sqrt{9-z^2} \text{ e } -3 \leq z \leq 3\right\} \\ &= \left\{(x,z) \in \mathbb{R}^2 : -\sqrt{9-z^2} \leq z \leq \sqrt{9-z^2} \text{ e } -3 \leq x \leq 3\right\}. \end{aligned}$$

Nessas condições, de acordo com o Teorema de Fubini, o volume do sólido é

$$\begin{aligned} \iiint_{\Omega} 1 \cdot \mathrm{d}V &= \iint_{Uxz} \left[\int_{-2}^{1-z} 1 \cdot \mathrm{d}y\right] \mathrm{d}A \\ &= \int_{-3}^{3} \int_{-\sqrt{9-z^2}}^{\sqrt{9-z^2}} \left[\int_{-2}^{1-z} 1 \cdot \mathrm{d}y\right] \mathrm{d}x \ \mathrm{d}z \\ &= \int_{-3}^{3} \int_{-\sqrt{9-x^2}}^{\sqrt{9-x^2}} \left[\int_{-2}^{1-z} 1 \cdot \mathrm{d}y\right] \mathrm{d}z \ \mathrm{d}x \\ &= 27\pi. \end{aligned}$$

Observe que $U_{xz} \subset \mathbb{R}^2$ possui simetria circular.

Isso sugere o uso de coordenadas cilíndricas no cálculo da integral tripla em $\Omega \subset \mathbb{R}^3$ definidas por

$$z = r\cos t, \quad y = y, \quad x = r\sin t.$$

Nesse caso, tem-se

$$\begin{aligned} U_{xz} &= U_{rt} \\ &= \{(t,r) \in \mathbb{R}^2 : 0 \leq r \leq 3 \text{ e } 0 \leq t \leq 2\pi\} \\ &= [0,3] \times [0,\pi]. \end{aligned}$$

Desse modo, pelo Teorema de Mudança de Variáveis, o volume do sólido é

$$\begin{aligned}\iiint_\Omega 1\cdot \mathrm{d}V &= \iint_{Urt}\left[\int_{-2}^{1-r\cos t} 1\cdot r\cdot \mathrm{d}y\right]\mathrm{d}A\\ &= \int_0^{2\pi}\int_0^3\left[\int_{-2}^{1-r\cos t} 1\cdot r\cdot \mathrm{d}y\right]\mathrm{d}r\ \mathrm{d}t\\ &= 27\pi.\end{aligned}$$

No Maple, as integrais iteradas acima são calculadas usando a seguinte sequência de comandos.

$\#\ Calculando\ o\ volume,\ com\ a\ projeção\ no\ plano\ coordenado\ yz.$
$\#\ Forma\ inerte.$
$Int(Int(Int(1,x=-sqrt(-z^2+9)..sqrt(-z^2+9)),y=-2..-z+1),z=-3..3);$
$\#\ Efetuando\ a\ conta.$
$int(int(int(1,x=-sqrt(-z^2+9)..sqrt(-z^2+9)),y=-2..-z+1),z=-3..3);$
$\#\ Calculando\ o\ volume,\ com\ a\ projeção\ no\ plano\ xz.$
$\#\ Forma\ inerte.$
$Int(Int(Int(1,y=-2..-z+1),x=-sqrt(9-z*z)..sqrt(9-z*z)),z=-3..3);$
$\#\ Efetuando\ a\ conta.$
$Int(r,y=-2..1-r*cos(t));$
$Int(Int(r,y=-2..1-r*cos(t)),r=0..3);$
$Int(Int(Int(r,y=-2..1-r*cos(t)),r=0..3),t=0..2*Pi);$
$\#\ Usando\ coordenadas\ polares\ na\ projeção\ do\ plano\ xz.$
$int(int(int(r,y=-2..1-r*cos(t)),r=0..3),t=0..2*Pi);$

Uma vez que os limites de integração das integrais repetidas da cunha $\Omega\subset\mathbb{R}^3$ estão fixados pela escolha de U_{yz} ou de U_{xz}, a integral tripla de uma função qualquer $f(x,y,z)$ em Ω é, de acordo com o Teorema de Fubini, dada por

$$\begin{aligned}\iiint_\Omega f(x,y,z)\ \mathrm{d}V &= \int_{-3}^{3}\int_{-\sqrt{9-z^2}}^{\sqrt{9-z^2}}\left[\int_{-2}^{1-z} f(x,y,z)\cdot \mathrm{d}y\right]\mathrm{d}x\ \mathrm{d}z\\ &= \int_{-3}^{3}\int_{-\sqrt{9-x^2}}^{\sqrt{9-x^2}}\left[\int_{-2}^{1-z} f(x,y,z)\cdot \mathrm{d}y\right]\mathrm{d}z\ \mathrm{d}x\\ &= \int_0^{2\pi}\int_0^3\left[\int_{-2}^{1-r\cos t} f(r\sin t,y,r\cos t)\cdot r\cdot \mathrm{d}y\right]\mathrm{d}r\ \mathrm{d}t.\end{aligned}$$

Do ponto de vista computacional, definida a função f, o cálculo da integral tripla é efetuado imediatamente pelo programa.

A título de ilustração, em coordenadas cartesianas, seja f a função dada por

$$f(x,y,z)=x^2-y+z^2.$$

Em coordenadas cilíndricas,

$$z=r\cos t,\quad y=y,\quad x=r\sin t,$$

tem-se

$$f(r\sin t, y, r\cos t) = r^2 - y.$$

Por conseguinte, decorre que

$$\begin{aligned}\iiint_\Omega f(x,y,z)\ \mathrm{d}V &= \int_0^{2\pi}\int_0^3\left[\int_{-2}^{1-r\cos t}(r^2-y)\cdot r\cdot \mathrm{d}y\right]\mathrm{d}r\ \mathrm{d}t\\ &= \frac{999\pi}{8}.\end{aligned}$$

No Maple, a sequência de comandos abaixo determina a integral tripla de f em Ω.

$\#\ Definindo\ a\ função\ dada.$
$f := (x,y,z)->x^2-y+z^2; f(x,y,z);$
$\#\ Introduzindo\ as\ novas\ variáveis.$
$x := (t,r,y)->r*sin(t); x(t,r,y);$
$z := (t,r,y)->r*cos(t); z(t,r,y);$
$\#\ Verificando\ a\ composição\ e\ eventual\ simplificação.$
$f(x(t,r,y),y,z(t,r,y));$
$simplify(f(x(t,r,y),y,z(t,r,y)));$
$\#\ Definindo\ a\ função\ composta\ nas\ novas\ variáveis.$
$F := (t,r,y)->simplify(f(x(t,r,y),y,z(t,r,y))); F(t,r,y);$
$\#\ Verificando\ a\ integral\ tripla\ de\ forma\ inerte.$
$Int(Int(Int(F(t,r,y)*r, y=-2..1-r*cos(t)), r=0..3), t=0..2*Pi);$
$\#\ Efetuando\ a\ integral\ tripla.$
$int(int(int(F(t,r,y)*r, y=-2..1-r*cos(t)), r=0..3), t=0..2*Pi);$

3.3 Uma cuia esférica com espaço interno cônico

Considere um sólido limitado abaixo pela esfera, em coordenadas esféricas *padronizadas*,

$$\rho = 2\sin\phi\cos\theta,$$

e acima pelo cone $x = \sqrt{y^2+z^2}$.

Inicialmente, observe que, das coordenadas esféricas *padronizadas*,

$$x = \rho\sin\phi\cos\theta, \quad y = \rho\sin\phi\sin\theta, \quad z = \rho\cos\phi,$$

segue que

$$\begin{aligned}\rho = 2\sin\phi\cos\theta &\Leftrightarrow \rho^2 = 2\rho\sin\phi\cos\theta\\ &\Leftrightarrow x^2+y^2+z^2 = 2x\\ &\Leftrightarrow x^2-2x+y^2+z^2 = 0\\ &\Leftrightarrow (x-1)^2+y^2+z^2 = 1.\end{aligned}$$

Assim, reconhece-se que a esfera dada tem centro $(1,0,0)$ e raio 1.

Inspirado pelas coordenadas esféricas, tem-se que uma parametrização *natural* dessa esfera é

$$X(u,v) = (1 + \sin u \cos v, \sin u \sin v, \cos u), \quad u \in [0,\pi]\,, \quad v \in [0, 2\pi]\,.$$

Por seu turno, uma parametrização *natural* do cone $x = \sqrt{y^2+z^2}$ é dada por

$$Y(u,v) = (\sqrt{u^2+v^2}, u, v), \quad u \in \mathbb{R}, \quad v \in \mathbb{R}.$$

No Maple, a sequência de comandos abaixo

$restart;$
$with(plots) : with(plottools) : with(Student) : with(ColorTools) : PaletteNames() :$
$\#Da\ esfera\ r = 2 * sin(phi) * cos(theta).$
$X := (u,v) - > [1 + sin(u) * cos(v), sin(u) * sin(v), cos(u)]; X(u,v);$
$esfera := plot3d(X(u,v), u = 0..Pi, v = 0..2 * Pi, scaling = constrained,$
$transparency = 0.2, axes = normal, style = surfacewireframe,$
$colour = v * cos(u), tickmarks = [[-3,3],[-3,3],[-3,3]], orientation = [55,61,4]);$
$\#\ Do\ cone\ x = sqrt(x * x + z * z).$
$Y := (u,v) - > [sqrt(u * u + v * v), u, v]; Y(u,v);$
$cne1 := plot3d(Y(u,v), u = -1.2..1.2, v = -1.2..1.2, scaling = constrained,$
$transparency = 0.3, axes = normal, style = surfacewireframe,$
$colour = "NiagaraDarkOrchid", tickmarks = [[-2,2],[-4,-3,3,4],[-3,-2,2,3]],$
$orientation = [55,61,4]);$
$\#\ Modelo\ da\ cuia.$
$display(cne1, esfera, scaling = constrained, axes = normal,$
$tickmarks = [[-3,3],[-2,4],[-3,3]], orientation = [-112,4,74]);$

gera o seguinte modelo



Fig. 14: Autoria própria

A esfera $(x-1)^2 + y^2 + z^2 = 1$ e o cone $x = \sqrt{y^2+z^2}$ intersectam-se, gerando uma curva C, tal que

$$(x-1)^2 + x^2 = 1 \quad \Leftrightarrow \quad x = 1 \text{ ou } x = 0.$$

Disso e da parametrização $Y(u,v)$, decorre que

$$u^2 + v^2 = 1.$$

Tomando $u = \cos t$ e $v = \sin t$, com $t \in [0, 2\pi]$, segue que uma parametrização de C é dada por

$$\sigma(t) = (1, \cos t, \sin t), \quad 0 \le t \le 2\pi.$$

Em consequência, referente ao cone, os excessos apresentados no modelo podem ser suprimidos tomando a parametrização

$$Y(r,t) = (r^2, r\cos t, r\sin t), \quad 0 \le r \le 1, \quad 0 \le t \le 2\pi.$$

Relativo à esfera, cuja parametrização é

$$X(u,v) = (1 + \sin u \cos v, \sin u \sin v, \cos u),$$

os excessos podem ser aparados tomando

$$u \in [0, \pi] \quad \text{e} \quad v \in \left[\frac{\pi}{2}, \frac{3\pi}{2}\right],$$

uma vez que

$$0 \le 1 + \sin u \cos v \le 1 \quad \Leftrightarrow \quad -1 \le \sin u \cos v \le 0$$

No Maple, a sequência de instruções

```
# Inteseção do cone e da esfera. Curva C.
sigma := t-> [1, cos(t), sin(t)]; sigma(t);
crv := plot3d(sigma(t), t = 0..2*Pi, colour = black, thickness = 2);
# Aparando os excessos.
# Cone.
u := (r,t)-> r*cos(t); v := (r,t)-> r*sin(t);
Y(u(r,t), v(r,t));
plot3d(Y(u(r,t), v(r,t)), r = 0..1, t = 0..2*Pi, scaling = constrained,
transparency = 0.3, axes = normal, style = surfacewireframe,
colour = "NiagaraDarkOrchid", tickmarks = [[-1, 1], [-2, -1, 1, 2], [-1, -2, 2, 1]],
orientation = [55, 61, 4]);
cne2 := plot3d(Y(u(r,t), v(r,t)), r = 0..1, t = 0..2*Pi, scaling = constrained,
transparency = 0.3, axes = normal, style = surfacewireframe,
colour = "NiagaraDarkOrchid",
tickmarks = [[-2, 2], [-4, -3, 3, 4], [-3, -2, 2, 3]], orientation = [55, 61, 4]);
# Esfera.
plot3d(X(u,v), u = 0..Pi, v = Pi/2..(3*Pi)/2, scaling = constrained,
transparency = 0.3, axes = normal, style = surfacewireframe,
colour = sin(v)*cos(u), tickmarks = [[-3, 3], [-3, 3], [-3, 3]],
orientation = [55, 61, 4]);
esfera2 := plot3d(X(u,v), u = 0..Pi, v = Pi/2..(3*Pi)/2,
scaling = constrained, transparency = 0.3, axes = normal,
style = surfacewireframe, colour = sin(v)*cos(u),
tickmarks = [[-3, 3], [-3, 3], [-3, 3]], orientation = [55, 61, 4]);
# Cuia aparada.
display(cne2, crv, esfera2, scaling = constrained,
axes = normal, tickmarks = [[-2, -1, 1, 2], [-2, -1, 1, 2], [-2, -1, 1, 2]],
orientation = [-177, -29, 38]);
```

gera a cuia esférica com espaço interno cônico abaixo.



Fig. 15: Autoria própria

Exame dinâmico, no Maple, da projeção do sólido sobre os planos coordenados revela que *não* é conveniente escolher projeção sobre o plano xy *nem* sobre o plano xz.



Fig. 16: Autoria própria



Fig. 17: Autoria própria

Seja $U_{yz} \subset \mathbb{R}^2$ a projeção do sólido $\Omega \subset \mathbb{R}^3$ sobre o plano coordenado $x = 0$. Por definição,

$$U_{yz} = \left\{(y, z) \in \mathbb{R}^2 : 0 \leq y^2 + z^2 \leq 1\right\}.$$

Em coordenadas cartesianas, do Teorema de Fubini, decorre que o volume do sólido é

$$\begin{aligned}\iiint_{\Omega} 1 \cdot \mathrm{d}V &= \iint_{U_{yz}} \left[\int_{1-\sqrt{1-y^2-z^2}}^{\sqrt{y^2+z^2}} 1 \cdot \mathrm{d}x\right] \mathrm{d}A \\ &= \int_{-1}^{1} \int_{-\sqrt{1-y^2}}^{\sqrt{1-y^2}} \left[\int_{1-\sqrt{1-y^2-z^2}}^{\sqrt{y^2+z^2}} 1 \cdot \mathrm{d}x\right] \mathrm{d}z\ \mathrm{d}y \\ &= \frac{\pi}{3}.\end{aligned}$$

Embora as integrais iteradas acima sejam resolvidas, computacionalmente, em poucos segundos, do ponto de vista manuscrito, elas demandam certo esforço braçal.

Isso motiva ou, pelo menos, deveria motivar, a pessoa a procurar por outros sistemas de coordenadas na resolução do problema.

Com efeito, o uso do sistema diferenciado de coordenadas *cilíndricas* definido por

$$x = x, \quad y = r\cos t, \quad z = r\sin t,$$

implica que o jacobiano da transformação é

$$\frac{\partial(x,y,z)}{\partial(x,t,r)} \quad = \quad -r,$$

e que a *nova* região de integração da integral dupla é dada por

$$U_{tr} = \left\{(t,r) \in \mathbb{R}^2 : 0 \le r \le 1 \text{ e } 0 \le t \le 2\pi\right\}.$$

Pelo Teorema de Mudança de Variáveis, nas coordenadas cilíndricas diferenciadas fornecidas, segue que o volume do sólido é

$$\begin{aligned}
\iiint_{\Omega} 1 \cdot \mathrm{d}V \quad &= \quad \iint_{U_{tr}} \left[\int_{1-\sqrt{1-r^2}}^{r} 1 \cdot r \cdot \mathrm{d}x\right] \mathrm{d}A \\
&= \quad \int_0^{2\pi}\int_0^1 \left[\int_{1-\sqrt{1-r^2}}^{r} 1 \cdot r \cdot \mathrm{d}x\right] \mathrm{d}r \; \mathrm{d}t \\
&= \quad \frac{\pi}{3}.
\end{aligned}$$

Do ponto de vista braçal, as integrais repetidas acima parecem mais agradáveis.

Acontece que a situação pode ser melhorada um pouco mais, com o uso do sistema diferenciado de coordenadas *esféricas* definido por

$$x = r\cos p, \quad y = r\sin p\cos v, \quad z = r\sin p\sin v.$$

Nessas condições, o jacobiano da transformação é

$$\frac{\partial(x,y,z)}{\partial(v,p,r)} \quad = \quad -r^2\sin p,$$

e a *nova* região de integração é dada por

$$U_{vp} = \left\{(v,p) \in \mathbb{R}^2 : 0 \le v \le 2\pi \text{ e } \frac{\pi}{4} \le p \le \frac{\pi}{2}\right\}.$$

Pelo Teorema de Mudança de Variáveis, nas coordenadas esféricas diferenciadas fornecidas, segue que o volume do sólido é

$$\begin{aligned}
\iiint_{\Omega} 1 \cdot \mathrm{d}V \quad &= \quad \iint_{U_{vp}} \left[\int_0^{2\cos p} 1 \cdot r^2\sin p \cdot \mathrm{d}r\right] \mathrm{d}A \\
&= \quad \int_0^{2\pi}\int_{\pi/4}^{\pi/2} \left[\int_0^{2\cos p} 1 \cdot r^2\sin p \cdot \mathrm{d}r\right] \mathrm{d}p \; \mathrm{d}v \\
&= \quad \frac{\pi}{3}.
\end{aligned}$$

No Maple, os cálculos acima foram realizados usando as seguintes intruções.

$$
\begin{aligned}
&\#\ Volume\ em\ coordenadas\ cartesianas.\\
&Int(1, x = 1 - sqrt(-y^2 - z^2 + 1)..sqrt(y^2 + z^2));\\
&Int(Int(1, x = 1 - sqrt(-y^2 - z^2 + 1)..sqrt(y^2 + z^2)),\\
&z = -sqrt(-x^2 + 1)..sqrt(-x^2 + 1));\\
&Int(Int(Int(1, x = 1 - sqrt(-y^2 - z^2 + 1)..sqrt(y^2 + z^2)),\\
&z = -sqrt(-y^2 + 1)..sqrt(-y^2 + 1)), y = -1..1);\\
&int(int(int(1, x = 1 - sqrt(-y^2 - z^2 + 1)..sqrt(y^2 + z^2)),\\
&z = -sqrt(-y^2 + 1)..sqrt(-y^2 + 1)), y = -1..1);\\
&\#\ Volume\ em\ coordenadas\ cilíndricas.\\
&Int(r, x = 1 - sqrt(-r^2 + 1)..r);\\
&Int(Int(r, x = 1 - sqrt(-r^2 + 1)..r), r = 0..1);\\
&Int(Int(Int(r, x = 1 - sqrt(-r^2 + 1)..r), r = 0..1), t = 0..2 * Pi);\\
&int(int(int(r, x = 1 - sqrt(-r^2 + 1)..r), r = 0..1), t = 0..2 * Pi);\\
&\#\ Volume\ em\ coordenas\ esféricas.\\
&Int(r^2 * sin(p), r = 0..2 * cos(p));\\
&Int(Int(r^2 * sin(p), r = 0..2 * cos(p)), p = Pi/4..Pi/2);\\
&Int(Int(Int(r^2 * sin(p), r = 0..2 * cos(p)), p = Pi/4..Pi/2), v = 0..2 * Pi);\\
&int(int(int(r^2 * sin(p), r = 0..2 * cos(p)), p = Pi/4..Pi/2), v = 0..2 * Pi);
\end{aligned}
$$

Uma vez que os limites de integração das integrais repetidas da cuia esférica com espaço interno cônico em $\Omega \subset \mathbb{R}^3$ estão fixados pela escolha de U_{yz} em coordenadas cartesianas, ou de U_{tr} em coordenadas cilíndricas, ou de U_{vp} em coordenadas esféricas, a integral tripla de uma função qualquer f em Ω é, de acordo com o Teorema de Fubini, dada por

$$
\begin{aligned}
V &= \iiint_\Omega f(x,y,z)\ \mathrm{d}V\\
&= \int_{-1}^{1}\int_{-\sqrt{1-y^2}}^{\sqrt{1-y^2}}\int_{1-\sqrt{1-y^2-z^2}}^{\sqrt{y^2+z^2}} f(x,y,z)\ \mathrm{d}x\ \mathrm{d}z\ \mathrm{d}y\\
&= \int_0^{2\pi}\int_0^1\int_{1-\sqrt{1-r^2}}^{r} r\cdot f(x, r\cos t, r\sin t)\ \mathrm{d}x\ \mathrm{d}r\ \mathrm{d}t\\
&= \int_0^{2\pi}\int_{\pi/4}^{\pi/2}\int_0^{2\cos p} r^2\sin p\cdot f(r\cos p, r\sin p\cos v, r\sin p\sin v)\ \mathrm{d}r\ \mathrm{d}p\ \mathrm{d}v.
\end{aligned}
$$

Do ponto de vista computacional, definida a função f, o cálculo da integral tripla é efetuado imediatamente pelo programa.

A título de ilustração, em coordenadas cartesianas, seja f a função dada por

$$f(x,y,z) = -x^2 + y^2 + z^2.$$

Em coordenadas esféricas,

$$x = r\cos p,\quad y = r\sin p\cos v,\quad z = r\sin p\sin v,$$

tem-se

$$f(r\cos p, r\sin p\cos v, r\sin p\sin v) = r^2 - 2r^2\cos^2 p.$$

Por conseguinte, decorre que

$$\begin{aligned} I &= \iiint_\Omega f(x,y,z)\ \mathrm{d}V \\ &= \int_0^{2\pi}\int_{\pi/4}^{\pi/2}\int_0^{2\cos p} r^2\sin p\cdot(r^2-2r^2\cos^2 p)\ \mathrm{d}r\ \mathrm{d}p\ \mathrm{d}v \\ &= \frac{\pi}{15}. \end{aligned}$$

No Maple, a sequência de comandos abaixo determina a integral tripla de f em Ω.

$\# \; Definindo \; a \; função \; dada.$
$f := (x,y,z)->(-1)*x*x+y*y+z*z; f(x,y,z);$
$\# \; Introduzindo \; as \; coordenadas \; esféricas \; diferenciadas.$
$x := (v,p,r)->r*cos(p); y := (v,p,r)->r*sin(p)*cos(v);$
$z := (v,p,r)->r*sin(p)*sin(v);$
$\# \; Verificando \; a \; composição \; e \; eventual \; simplificação.$
$f(x(v,p,r),y(v,p,r),z(v,p,r)); simplify(f(x(v,p,r),y(v,p,r),z(v,p,r)));$
$\# \; Definindo \; a \; função \; composta \; nas \; novas \; variáveis.$
$F := (v,p,r)->simplify(f(x(v,p,r),y(v,p,r),z(v,p,r))); F(v,p,r);$
$\# \; Verificando \; a \; integral \; tripla \; de \; forma \; inerte.$
$Int(Int(Int(r^2*sin(p)*F(v,p,r),r=0..2*cos(p)),p=Pi/4..Pi/2),v=0..2*Pi);$
$\# \; Efetuando \; a \; integral \; tripla.$
$int(int(int(r^2*sin(p)*F(v,p,r),r=0..2*cos(p)),p=Pi/4..Pi/2),v=0..2*Pi);$

4 CONCLUSÃO

Nesta oficina, a partir de problemas propostos, desenvolveu-se, de forma interativa, a resolução de integrais triplas por meio do Teorema de Fubini, conforme delineado nas duas primeiras seções iniciais. Primeiro, com o Maple, construiu-se o sólido $\Omega \subset \mathbb{R}^3$. Após, examinou-se a situação, decidindo que projeção e curva são mais convenientes. Por fim, instruiu-se o programa a efetuar ordenadamente as integrais repetidas pertinentes.

Nesse sentido, espera-se que a presente oficina, embora de caráter elementar, sirva de estímulo e incentivo aos participantes para que usem o Maple como ferramenta de auxílio no processo de aprendizagem; particularmente, no estudo de integrais triplas.

Bibliografia

[1] ANDRADE, L. N. **Introdução à Computação Algébrica com o Maple**. Rio de Janeiro: SBM, 2004.

[2] ROGAWSKI, J.; ADAMS, C. **Cálculo**, 3. ed. Porto Alegre: Bookman, 2018. v. 2.

[3] WEIR, M. D.; HASS, J.; GIORDANO, F. R. **Cálculo (George B. Thomas Jr)**, 11. ed. São Paulo: Addison Wesley, 2009. v. 2.